وَمَا أُوتِيتُمْ مِنَ الْعِلْمِ إِلَّا قَلِيلًا

Dr. Hesham Omran
Ain Shams University – Master Micro LLC

Analog IC Design (Xschem, Ngspice, ADT)
Lab 06
Differential Amplifier

Intended Learning Objectives

In this lab you will:

- Design a differential amplifier circuit using the Sizing Assistant (SA).
- Learn how to simulate the small-signal differential characteristics of a differential amplifier.
- Learn how to simulate the small-signal common-mode characteristics of a differential amplifier.
- Learn how to simulate the large-signal differential characteristics of a differential amplifier.
- Learn how to simulate the large-signal common-mode characteristics of a differential amplifier.

Part 1: Differential Amplifier Design

1) We want to design a resistive loaded differential amplifier with the specifications below.

Parameter	
Supply (V_{DD})	$1.8V$
Current seed (Ref current)	$20\mu A$
Bias current (I_{SS})	$40\mu A$
Differential gain	8
CM output level[1]	$V_{DD}/3$
Load capacitance	$1pF$

➔ **Note:** that the bias current is split between two transistors; each transistor gets $I_D = 20\mu A$.

2) Since the required output level is closer to the ground rail, we will use a PMOS input stage. Assume the PMOS will not be placed in a dedicated well to save the area. Assume we will use a simple current mirror for biasing.
3) The design schematic is shown below. Note that:

[1] The supply voltage is divided between the tail current source, the input stage, and the resistive load. The tail current source needs a good voltage drop (V_{DS}) for higher output resistance and good CMRR. The input stage needs a good voltage drop (V_{DS}) for higher output resistance and to allow good signal swing. The resistive load needs a good voltage drop to allow a good swing and to give higher gain. Note that $I_{SS}R_D = 2V_{out-CM}$ must be smaller than ($V_{DD} - V_{dsat-tail}$) for proper large signal characteristics (why?).

a. We will create the circuit under test in a schematic, create a symbol, then create a new schematic for the testbench.
b. We added input pins (ipin), output pins (opin) and input/output pins (iopin) that will define the symbol pins.
c. We used variables to define the component parameters. We will pass the variables to the upper level of hierarchy.



4) Create a symbol from the menu bar: Symbol -> Make symbol from schematic.
5) Edit the symbol to look neat and nice. In an empty place press ‘q’ and define the parameters in the symbol as shown below. For more information about parameters, check this: http://repo.hu/projects/xschem/xschem_man/parameters.html.

```
type=subcircuit
format="@name @pinlist @symname WP1=@WP1 LP1=@LP1 WP2=@WP2 LP2=@LP2 RD=@RD"
template="name=x1 WP1=1u LP1=0.5u WP2=1u LP2=0.5u RD=1k"
```



6) **Choose** $\boldsymbol{R_D}$ to meet the CM output level spec.
7) The differential amplifier gain is given by

$$|A_v| \approx g_m(R_D||r_o)$$

8) We will choose L to set $r_o \gg R_D \rightarrow r_o = 10 \times R_D$

$$r_o = 10 \times \frac{V_{R_D}}{R_D}$$

$$|A_v| \approx 0.91 \times g_m R_D = 0.91 \times \frac{g_m}{I_D} \times V_{R_D}$$

9) **Choose g_m/I_D** to meet the differential gain spec.

$$\frac{g_m}{I_D} = \frac{|A_v|}{0.91 \times V_{R_D}}$$

10) Assume we will set V_{DS} of the tail current source to $300mV$ to allow more output swing. **Report the input pair sizing and gm/ID using SA.**



11) Given the above assumption for V_{DS} of the tail current source, **calculate** the required CM input level.

12) For the tail current source, use the following specifications:

Parameter	
Input current	$20\mu A$
Percent mismatch: $\boldsymbol{\sigma(I_{out})/I_{out}}$	$\leq 2\%$
Lambda	0.1
Compliance voltage	$\leq 200mV$
Area	Minimize

13) Use SA to plot the sizing at a constant $\boldsymbol{\sigma(I_{out})}/I_{out}$ which is given by

$$\frac{\boldsymbol{\sigma(I_{out})}}{\boldsymbol{I_{out}}} = \frac{\sqrt{\mathbf{1} + \frac{\mathbf{1}}{\boldsymbol{m}} \times \boldsymbol{idmis}}}{I_D} \times \mathbf{100}$$

The $\sqrt{1 + \frac{1}{m}}$ factor is due to taking into account the effect of two random variables (V_{TH} of the two current mirror transistors).

Note: idmis and ID are those of the 20uA device (the reference device) and the 40uA device (the mirror output device) variance is idmis^2/m, where m=2 because it has twice the area. Note that when we add two uncorrelated random variables, we add the variances, not the standard deviation.

Plot L, W, AREA, and λ at the settings shown below. The results will not be smooth due to the noisy MC mismatch data in the LUT.



14) As seen in the plot above, for a given mismatch requirement, the minimum area is achieved at the min gm/ID (largest V*). Similarly, for a given area requirement, the minimum mismatch is achieved at the min gm/ID (largest V*). That's why current mirrors are commonly biased in strong inversion.
15) The plot also shows that at a given V^* (compliance voltage), going for lower mismatch necessitates longer L, which also gives lower λ (higher r_o, although the effect of L on λ is limited at low V_{DS}). However, this comes at the expense of area.
16) In order to satisfy the both mismatch and lambda requirements, which design curve will you pick, the 1% or the 2% mismatch?
17) Given the compliance voltage spec, **report** the above figure with a cursor added to the selected design point.
18) **Calculate** the min and max CM input levels. Is the previously selected CM input level in the valid range?

Part 2: Differential Amplifier Simulation

1) Create a new cell for the testbench. Create the testbench schematic as shown below.



➔ Hint: Use asrc at the input and output instead of the baluns to combine and separate common-mode and differential signals.



2) Set the transistor sizing and Vicm as designed in Part 1.

Report the following:

1) OP simulation.

- Report the schematic of the diff pair with DC OP point clearly annotated.

ID
VGS
VDS
VTH
VDSAT
Vstar = 2/(gm/ID)
gm/ID
GM
GDS
GMB
Region

- Check that all transistors operate in saturation.

➔ Hint: You can edit the transistor symbol to add gm/ID OP annotation.

```
tcleval(gmoverid=[to_eng [expr [ngspice::get_node [subst -nocommand
{\@m.${path}@spiceprefix@name\.m0\[gm]}]] / [ngspice::get_node
[subst -nocommand {i(\@m.${path}@spiceprefix@name\.m0\[id])}]]]]])
```

2) Diff small signal ccs:

- Use AC magnitude = 1 for the diff source (and AC magnitude = 0 for the CM source).
- Run AC analysis (1Hz:10GHz, logarithmic, 10 points/decade).
- Report the Bode plot of small signal diff gain.
- Compare the DC diff gain and BW with hand analysis in a table.

➔ **Hint: In the waveform graph, you can plot db20 and unwrapped phased using the expressions below:**

```
Av DIFF dB; vod db20()
Av DIFF Phase; ph(vod) cph()
```

3) CM small signal ccs:

- Use AC magnitude = 1 for the CM source (and AC magnitude = 0 for the diff source).
- Run AC analysis (1Hz:10GHz, logarithmic, 10 points/decade).
- Report the Bode plot of small signal CM gain.
- Compare the DC CM gain with hand analysis in a table. Is it smaller than “1”? Why?
- Justify the variation of Avcm vs frequency.
- Plot Avd/Avcm in dB. Compare Avd/Avcm @ DC with hand analysis in a table.
- Justify the variation of Avd/Avcm with frequency.

➔ **Example ngspice code for parts 1 to 3:**

```
.param IB=20u CL=1p VICM_DC={1.8-0.3-0.94}
.control

alterparam sw_stat_global = 0
alterparam sw_stat_mismatch = 0
reset

save all

let x = 1
set num = {$&x}
dowhile x <= 4
save @m.x1.xm{$num}.m0[id]
save @m.x1.xm{$num}.m0[vgs]
save @m.x1.xm{$num}.m0[vds]
save @m.x1.xm{$num}.m0[vdsat]
save @m.x1.xm{$num}.m0[vth]
save @m.x1.xm{$num}.m0[gm]
save @m.x1.xm{$num}.m0[gds]
save @m.x1.xm{$num}.m0[gmbs]
save @m.x1.xm{$num}.m0[cgg]
save @m.x1.xm{$num}.m0[cdd]
save @m.x1.xm{$num}.m0[css]
let x = x + 1
set num = {$&x}
```

```
end

* === OP Point ===
op
write diff_amp_tb_op_ac.raw
set appendwrite

echo \"        === MOSFET Operating Points ===\"
show m : id : vgs : vds : vth : vdsat : gm : gds : gmbs

echo \"        === Bias Points ===\"
let x = 1
dowhile x <= 4
  set num = {$&x}
  let gmid = @m.x1.xm{$num}.m0[gm]/@m.x1.xm{$num}.m0[id]
  let Vstar = 2/gmid
  echo \"M{$num}: gm/id = $&gmid S/A, Vstar = $&Vstar V\"
  let x = x + 1
end

* === DIFFERENTIAL AC ===
alter V2 AC = 1
alter V4 AC = 0
ac dec 10 1 10e9

* === COMMON MODE AC ===
alter V2 AC = 0
alter V4 AC = 1
ac dec 10 1 10e9

* === EXTRACT RESULTS ===
set units=degrees
let vod = ac1.v(VODIFF)
let vocm = ac2.v(VOCM)
let CMRR = vdb(vod) - vdb(vocm)

* === MEASUREMENTS ===
meas ac Av_diff_db find vdb(vod) at=1
meas ac Av_diff_mag find vmag(vod) at=1
let A3db = Av_diff_mag/sqrt(2)
meas ac BW_diff when vmag(vod)=A3db fall=1

meas ac Av_CM_db find vdb(vocm) at=1
meas ac Av_CM_mag find vmag(vocm) at=1
let A3db = Av_CM_mag/sqrt(2)
meas ac BW_CM when vmag(vocm)=A3db fall=1

let CMRR_db = Av_diff_db - Av_CM_db

* === DISPLAY RESULTS ===
echo \"        === AC Results ===\"
```

```
echo \"Av Diff: $&Av_diff_db dB, BW: $&BW_diff Hz\"
echo \"Av CM: $&Av_CM_db dB, BW: $&BW_CM Hz\"
echo \"CMRR: $&CMRR_db dB\"

write diff_amp_tb_op_ac.raw

.endc
```

4) Diff large signal ccs:

- Use dc sweep (not parametric sweep) for Vid = -VDD:10m:VDD.
- Report diff large signal ccs (VODIFF vs VIDIFF). Compare the extreme values with hand analysis in a table.
- Plot the deriv() of VODIFF vs VIDIFF. What is the peak value? Why?

➔ **Example ngspice code:**

```
.param IB=20u CL=1p VICM_DC={1.8-0.3-0.94}
.control

alterparam sw_stat_global = 0
alterparam sw_stat_mismatch = 0
reset

save all

* === DIFFERENTIAL LARGE SIGNAL ===
dc V2 -1.8 1.8 10m

* === EXTRACT EXTREME VALUES ===
let Vmax = maximum(VODIFF)
let Vmin = minimum(VODIFF)

* === DISPLAY RESULTS ===
echo \"Diff Large Signal Analysis Results:\"
echo \"Max Output: $&Vmax V\"
echo \"Min Output: $&Vmin V\"

write diff_amp_tb_dc_diff.raw

.endc
```

5) CM large signal ccs (region vs VICM):

- To plot the operating region of each transistor we can plot VDS and Vdsat vs VICM.
- Use DC sweep (not parametric sweep) for VICM = -1:10m:VDD (no need to run AC analysis). Note that we start the sweep from a negative number because the min VICM will turn out to be negative.
- Plot VDS and Vdsat parameter vs Vicm for the input pair and the tail current source.
- Find the CM input range (CMIR). Compare with hand analysis in a table.
- The min VICM is not as expected, why?



➔ **Example ngspice code:**

```
.param IB=20u CL=1p VICM_DC={1.8-0.3-0.94}
.control

alterparam sw_stat_global = 0
alterparam sw_stat_mismatch = 0
reset

save all

let x = 1
set num = {$&x}
dowhile x <= 4
save @m.x1.xm{$num}.m0[vds]
save @m.x1.xm{$num}.m0[vdsat]
let x = x + 1
set num = {$&x}
end

* === CM LARGE SIGNAL ===
dc V4 -1 1.8 10m

write diff_amp_tb_dc_cm.raw

.endc
```

6) CM large signal ccs (GBW vs Vicm):

- What we usually care about is the overall specs of the circuit and not the VDS of individual transistors (we cannot practically measure the VDS of every transistor in the lab).
- We will examine the variation of GBW vs Vicm. The bandwidth of this circuit is determined by RD and CL, i.e., it is independent of Vicm, thus Avd variation is itself GBW variation.

- Set VIDIFF AC to 1 (because we will measure Avd) and VICM DC and AC to 0 (we will sweep the VICM DC in the control block).



- Use AC analysis (start = 1, stop = 10, pts = 1) to get Avd. Use parametric sweep (not dc sweep) for Vicm = -1:50m:VDD.
- Report CM large signal ccs (Avd vs Vicm). Assume the valid range for Vicm (CMIR) is defined by the condition that Avd is within 90% of the max gain, i.e., 10% drop in gain.
- Find the CM input range. Compare with the previous method (Vdsat) in a table.

➔ **Example ngspice code:**

```
.param IB=20u CL=1p
.control
alterparam sw_stat_global = 0
alterparam sw_stat_mismatch = 0
reset
save all
let index = 0
compose VICM_vec start=-1 stop=1.8 step=50m
let Avd_vec = VICM_vec
setscale VICM_vec
foreach VICM_val $&VICM_vec
    alter V4 DC=$VICM_val
    ac dec 1 1 10
    meas ac Avd MAX vmag(vodiff) FROM=1 TO=10
    let Avd_vec[index] = Avd
    let index = index + 1
end

write diff_amp_tb_cmir.raw Avd_vec

plot Avd_vec vs VICM_vec
```

```
.endc
```

➔ **Hint:** In addition to ngspice plotting, you can plot the parametric sweep results using the waveform graph as shown below.



Lab Summary

- In Part 1 you learned:
  - How to design a resistive-loaded differential amplifier.
- In Part 2 you learned:
  - How to simulate the small-signal differential gain of a differential amplifier.
  - How to simulate the small-signal common-mode gain of a differential amplifier.
  - How to simulate the large-signal differential characteristics of a differential amplifier.
  - How to simulate the large-signal common-mode characteristics of a differential amplifier.

Acknowledgements

Thanks to all who contributed to these labs. Special thanks to Dr. Sameh A. Ibrahim for reviewing and editing the labs. If you find any errors or have suggestions concerning these labs, please contact Hesham.omran@eng.asu.edu.eg.